

Automating Code Generation for Higher Inductive Types

Paventhan Vivekanandan
School of Informatics, Computing and Engineering
Indiana University Bloomington
USA
pvivekan@indiana.edu

David Christiansen
School of Informatics, Computing and Engineering
Indiana University Bloomington
USA
davidchr@indiana.edu

Abstract

This paper provides a sample of a \LaTeX document which conforms, somewhat loosely, to the formatting guidelines for ACM SIG Proceedings.

Keywords ACM proceedings, \LaTeX , text tagging

1 Introduction

Type theory, a foundational language of mathematics, is a functional programming language and a theorem prover. We can view a type in type theory as a topological space in homotopy theory or a groupoid in a category theory. Elements of a type corresponds to points in a topological space and elements of identity type corresponds to paths in a topological space. Homotopy type theory extends Martin-Löf's intensional type theory by adding the notion of higher inductive type and univalence axiom. A higher inductive type is a general schema for defining new types using constructors for points and paths. The paths specified by a higher inductive type are realized by equivalences in the universe, and the mapping of paths to equivalences is made possible by univalence axiom [7].

Proof assistants like Coq [6] and Agda [5] does not have built-in mechanisms to support the specification of higher inductive types. However, we can work with higher inductive types in Coq and Agda by postulating them using axioms. In this approach, the elimination rule acting on the higher inductive type does not compute. Dan Licata suggested an alternative method to work with higher inductive types using type abstraction [4]. It involves defining a higher inductive type using an abstract type inside a module. The module consists of a boiler-plate code segment which defines the higher inductive type using a private base type. Inside the module, the recursion and the induction principles acts on the constructors of the private base type. The abstract type is then exported allowing the reduction rules for point constructors to hold definitionally.

Agda is a proof-assistant based on Martin-Löf's intensional type theory. In Agda, a metaprogram, which is a

programs used to construct other programs, runs in a built-in type checking monad. Inspired by Idris's elaborator [3], Agda's reflection library expose an interface to the type checker using the type checking monad which allows for the construction of more interesting metaprograms. The type checking monad provides the interface to the Agda type checker through a set of primitive operations which can be used to retrieve static type information regarding various code segments. The primitive operations can also be used to build code fragments using constructs of abstract syntax tree, and to convert an abstract syntax tree to its concrete syntax.

In this paper, we discuss a tool used to automate the generation of boiler-plate code segments of the higher inductive type and its reduction rules. This tool extensively uses the Agda reflection library and it builds the abstract syntax tree of the code segments with static type informations obtained using the reflection primitives. The generated code is then brought into scope by another top-level reflection primitive. More specifically, we discuss the following contributions.

- We introduce a tool to automate construction of the recursion and the induction principles for inductive types with constructors taking zero arguments, one or more arguments, and the type being defined itself as an argument.
- We discuss the automation of code generation of boiler-plate code for a higher inductive type defined inside a module. The boiler-plate code depends on a base type defined as private inside the module. The constructors of the private base type are not accessible outside the module.
- We discuss the automation of code generation of the elimination and the computation rules for a higher inductive type with point and path constructors. We demonstrate the code generation of the reduction rules, specified as postulates, for the path constructors of the higher inductive type.
- We discuss the automation of code generation for patch theory [2] implementation enriched with patches of encryption.

We automated the code generation of approximately 1500 lines of code of the patch theory implementation with just

70 lines of automation code. The automation code extensively uses the static type information queried from the type checker using the reflection primitives.

2 Background

2.1 Higher Inductive Type

In homotopy type theory, an element of an *identity type* is used to model the notion of a path in a topological space or a morphism in a groupoid. An element of an identity type $a =_A b$ states that a and b are propositionally equal. In type theory, *propositional equality* is a proof-relevant notion of equality, internal to the theory, expressed by identity types. We also have a proof-irrelevant notion of equality, external to the theory, known as *judgmental equality* or *definitional equality*. Definitional equality is used to express equality by definition. For instance, when we have a function $f : \text{Nat} \rightarrow \text{Nat}$ defined as $f(x) = x^2$ then the expression $f(3)$ is definitionally equal to 3^2 .

A higher inductive type extends normal inductive type by providing constructors for generating paths in addition to providing constructors for generating points. For example, consider the following higher inductive type definition of circle.

```
1
2 data Circle : Set where
3   base : Circle      -- point constructor
4   loop : base ≡ base  -- path constructor
5
```

In Agda and Coq, we don't have built-in primitives to support the definition of Circle. However, the higher inductive types including Circle can be specified using Dan Licata's method [4]. According to this method, a higher inductive type is defined using type abstraction. For example, Circle is defined using Dan Licata's method as follows.

```
1
2 module Circle where
3   private
4     data S* : Set where
5       base* : S*
6
7   S : Set
8   S = S*
9
10  base : S
11  base = base*
12
13  postulate
14    loop : base ≡ base
15
```

Inside the module Circle, the type S is defined using a private type S*. The constructor base is defined using base* and the path loop is given as a propositional equality. The recursion and the induction principles are defined to pattern match on the constructor base* of the type S*. The clients of Circle will not have access to the constructor base* of the private type S* as it is not visible outside the module.

The clients only way of access to the constructor is through the non-dependent and dependent elimination rules. The following code gives the non-dependent eliminator or the recursion rule recS.

```
1
2 recS : {C : Set} →
3   (cbase : C) →
4   (clloop : cbase ≡ cbase) →
5   S → C
6 recS cbase clloop base* = cbase
7
8 postulate
9   βrecS : {C : Set} →
10    (cbase : C) →
11    (clloop : cbase ≡ cbase) →
12    ap (recS cbase clloop) loop ≡ clloop
13
```

recS ignores the path argument and simply computes to the appropriate answer for the point constructor. The computational behavior for the path constructor loop is postulated using reduction rule β_{recS} . In the above code, the function ap gives the action of a function f on paths. The following equation gives the type of ap.

$$ap_f : (x =_A y) \rightarrow (f(x) =_A f(y)) \quad (1)$$

The following code gives the dependent eliminator or the induction rule indS and its computational rules.

```
1
2 indS : {C : S → Set} →
3   (cbase : C base) →
4   (clloop : transport C loop cbase ≡ cbase) →
5   (circle : S) → C circle
6 indS cbase clloop base* = cbase
7
8 postulate
9   βindS : {C : S → Set} →
10    (cbase : C base) →
11    (clloop : transport C loop cbase ≡ cbase) →
12    apd (indS {C} cbase clloop) loop ≡ clloop
13
```

In the above code, the function apd gives the action of dependent functions of type $f : \prod_{(x:A)} B(x)$ on paths. The following equation gives the type of apd.

$$apd_f : \prod_{p:x=y} (p_*(f(x)) =_{B(y)} f(y)) \quad (2)$$

In (2), the point $p_*(f(x))$ lying in space $B(y)$ can be thought of as an endpoint of a path obtained by lifting the path p from $f(x)$ to a path in the total space $\sum_{(x:A)} B(x) \rightarrow A$. The following equation gives the type of p_* which is also known as *transport* [7].

$$\text{transport}_p^B : B(x) \rightarrow B(y) \quad (3)$$

where $p : x = y$ for $x, y : A$.

2.2 Agda Reflection

Agda has a reflection library that enables compile time meta-programming. The reflection library provides an interface

to the Agda type checker through a built-in type checking TC monad. Using the reflection library and its meta-programming interface, we can convert a code fragment to its corresponding abstract syntax tree, and we can also convert the abstract representation back to its concrete agda code. Internally, agda translates the reflection code to a unique reflection syntax. The core library of Agda is enriched with functions supporting translations from reflection syntax to abstract syntax and from abstract syntax to concrete syntax and vice-versa. In this section, we will see usage examples for some of the primitive operations specified in the Agda reflection library which are used by the tool described in this paper. More information on the reflection library can be found in the Agda documentation [1].

In Agda, macros are used to construct a metaprogram that executes in a `Term` position. Macros are functions of type `t1 → t2 → ... → Term → TC T` defined inside a macro block. During a macro invocation, the last argument of type `Term` is provided by the type checker, and it represents the metavariable instantiated with the result of the macro invocation. Also, the arguments of type `Term` and `Name` are automatically quoted. For example, consider the following macro invocation on an identity function.

```

1
2 macro
3   mc1 : Term → Term → TC T
4   mc1 exp hole = bindTC (quoteTC exp)
5                   (λ exp' → unify hole exp')
6
7 sampleTerm : Term
8 sampleTerm = mc1 (λ (n : Nat) → n)
9

```

The value of `sampleTerm` is as follows:

$$\text{lam visible (abs "n" (var 0 []))} \quad (4)$$

In (4), the constructor `lam` encodes the abstract representation of the identity function. The argument `n` is represented as explicit and the body of the lambda function refers to `n` using de-bruijn index.

`Term` is the central type that defines an abstract syntax tree representation for the Agda terms. It is mapped to the `AGDATERM` built-in. The surface syntax for reflection primitives is not fixed, and so they are mapped to their corresponding built-ins, using `BUILTIN` pragma, to inform the type checker about the reflection concept being defined. In the above macro, the `quoteTC` primitive converts a concrete agda syntax into its `Term` representation. `unify` performs the unification of two terms and attempts to solve any meta variables present during the unification process. `bindTC` unwraps the result of `(quoteTC exp)`, which is captured in `exp'`, and passes it to `unify` for the unification process.

The `unquoteTC` primitive converts the abstract syntax tree representation of `Term` into its concrete Agda syntax. The following code gives an example usage for the `unquoteTC` primitive.

```

1
2 macro
3   mc2 : Term → Term → TC T
4   mc2 exp hole = bindTC (unquoteTC exp)
5                   (λ exp' → unify hole exp')
6
7 sampleSyntax : Nat → Nat
8 sampleSyntax = mc2 (lam visible (abs "n" (var 0 [])))
9

```

Macro `mc2` converts the abstract representation of the identity function given by (1) to its concrete syntax. We also have a type `Name`, an internal identifier mapped to a built-in type `QName`. In Agda, `Name` literals are created using `quote` primitive.

Metaprograms to create top-level definitions are brought into scope and executed using `unquoteDecl` or `unquoteDef` primitive. A function of a given type is declared using `declareDef`, and is later defined by `defineFun`. A postulate of a given type is defined using `declarePostulate`. Newly defined functions and postulates are brought into scope and executed by `unquoteDecl`. The usage of `declareDef`, `defineFun` and `unquoteDecl` is best illustrated with the following example:

```

1
2 plus : Nat → Nat → Nat
3 plus zero b = b
4 plus (suc n) b = suc (plus n b)
5

```

The above function computes addition of two natural numbers. It can be defined using Agda reflection primitives as follows:

```

1
2 pattern vArg x = arg (arg-info visible relevant) x
3 pattern _⇒_ a b = pi (vArg a) (abs "_" b)
4 pattern `Nat = def (quote Nat) []
5
6 unquoteDecl plus =
7   bindTC (declareDef (vArg plus) (`Nat `⇒ `Nat `⇒ `Nat)) λ _ →
8     defineFun plus
9       (clause (vArg (con (quote zero) []) ::
10                  vArg (var "y") :: [])
11              (var 0 []) ::
12              clause (vArg (con (quote suc)
13                             (vArg (var "x") :: [])) ::
14                  vArg (var "y") :: [])
15                    (con (quote suc)
16                      (vArg
17                        (def plus
18                          (vArg (var 1 []) ::
19                            vArg (var 0 []) :: [])) :: [])) :: [])
20

```

In the above code, `declareDef` declares the type of `plus` using the constructor `pi`, and `defineFun` uses the declared type information to define the clauses for `plus` using constructors `con` and `def`. Finally, the function `plus` is brought into scope by `unquoteDecl`. Terms use the de-bruijn index to refer to the declared variables. For example, the variable `y` declared at line 10 is referenced using a de-bruijn index `0`

at line 11. `vArg` encodes the visibility and relevance informations of the arguments to the reflection constructs.

In the next section, we will see automation of code generation for normal inductive types. In section 4, we will see automation of code generation for higher inductive types, and in section 5, we will revisit patch theory [2] application and discuss automation of code generation for the higher inductive types defined to model Darc's version control system.

3 Code Generation for Inductive Type

An inductive type X is a type that is freely generated by a finite collection of constructors. The constructors of X are functions with zero or more arguments and codomain X . The constructors can also take an element of type X itself as an argument, but only strictly positively. In Agda reflection library, data-type of type `Definition` stores the constructors of an inductive type as a list of `Name`. The type of a constructor can be retrieved by giving its `Name` as an input to the `getType` primitive. In the following subsections, we will discuss how to use the constructor informations to generate code for the elimination rules of an inductive type.

3.1 Non-dependent Eliminator

For an inductive type W , with a constructor g , the recursion principle says that to define a mapping $f : W \rightarrow P$, it suffices to define the action of f on input g . For example, the recursion principle for `Nat` says that a mapping $f : \text{Nat} \rightarrow P$ can be given by defining the action of f on the constructors `zero` and `(suc x)` for $x : \text{Nat}$.

The constructor g can take zero or more arguments including an element of the type W . Lets define the type of g as follows.

$$g : (A \rightarrow W) \rightarrow B \rightarrow W \rightarrow W$$

To define the action of f on input g , we need a function d of the following type.

$$d : (A \rightarrow W) \rightarrow (A \rightarrow P) \rightarrow B \rightarrow W \rightarrow P \rightarrow P$$

The function f will map the constructor g to d . To construct the type of the recursion rule for W , we need to build the type of d . We can retrieve the static type information of g using reflection primitives, and use that to construct the type of d . The constructor `pi` of type `Term` encodes the abstract syntax tree (AST) representation of g (fig. 1). We can retrieve and traverse the AST of g , and add new type information into it to build a new type representing d .

During the traversal of the abstract syntax tree of g , when we identify a function f_1 with codomain W , we add a new function f_2 , with the same arguments as f_1 and codomain P , to the tree. For example, in figure 1, a new function is built from the first argument $A \rightarrow W$ by modifying the codomain W to P . The new function of type $A \rightarrow P$ is added after the function $A \rightarrow W$ as in figure 2. Constant types require

```
(pi (vArg
  (pi (vArg `A) (abs "-" `W))) -- f1 : A → W
  (abs "-" (pi (vArg `B)      -- B
    (abs "-" (pi (vArg `W)    -- W
      (abs "-" `W))))))      -- W
```

Figure 1. Abstract syntax tree for constructor g

```
(pi (vArg
  (pi (vArg `A) (abs "-" `W))) -- f1 : A → W
  (abs "-" (pi (vArg
    (pi (vArg `A) (abs "-" `P))) -- f2 : A → P
    (abs "-" (pi (vArg `B)      -- B
      (abs "-" (pi (vArg `W)    -- W
        (abs "-" (pi (vArg `P)  -- P
          (abs "-" `P))))))))) -- P
```

Figure 2. Abstract syntax tree for constructor d

no modifications. Therefore, we copy B into the new type without any changes. When the type W occurs directly in a non-codomain position, we add the type P next to it. Finally, we change the codomain W of g to P resulting in an abstract syntax tree representation of d (fig. 2). We repeat this process for all the constructors of a given type. In the case of `Nat`, we construct the output types for `zero` and `(suc x)` for $x : \text{Nat}$ using the above method.

The parameter and the index of W , if present, should be encoded as part of the type of function f . We can retrieve information about the parameter and the index of W from its type. The constructors refer to the parameter and the index using de-bruijn indices. During the construction of the output type d , we should update the de-bruijn indices accordingly. The constructor data-type contains the count cp of parameters occurring in a defined type. It also encodes the constructors of the type as a list of `Name`. We can retrieve the index count by finding the difference between cp and length of the constructor list. The parameters are common to all the constructors of a type. But the index values are different for each constructor. So we have to encode unique indices for each constructor. Also, some constructors might not take the same number of indices as the parent type. For example, in the case of `Vec`, the constructor `[]` excludes the index `Nat` from its type. We do not have any reflection primitive to retrieve the index count from a constructor name. A workaround is to pass the index count of each constructor explicitly to the automation tool.

Once we have the type of d , we can build the type of the recursion rule for W . To encode the mapping $W \rightarrow P$ in the recursion type we need to declare P . We can use the constructor `agda-sort` to introduce the type $(P : \text{Set})$. The

```

(clause
  (vArg (var "P") ::                -- P
   vArg (var "d") ::                -- d
   vArg (con (quote g)
     (vArg (var "alpha") ::         -- α
      vArg (var "beta") ::         -- β
      vArg (var "omega") :: [])) :: [] -- ω
  (var 3
    (vArg (var 2 [] ::              -- α
     vArg (lam visible (abs "a"
       (def f
         (vArg (var 3
           (vArg (var 0 [] :: []) :: [])) :: [])) :: []
         vArg (var 1 [] ::          -- β
          vArg (var 0 [] ::          -- ω
           vArg (def f (vArg (var 0 [] :: []))))) :: f(ω)
       )
     )
  )

```

Figure 3. Clause definition for the computation rule of W

type of the recursion rule f is given as follows.

$$\begin{aligned}
 f : (P : \text{Set}) \rightarrow \\
 (d : (A \rightarrow W) \rightarrow (A \rightarrow P) \rightarrow B \rightarrow W \rightarrow P \rightarrow P) \rightarrow \\
 W \rightarrow P
 \end{aligned}$$

The above type is declared using `declareDef`. We can build the computation rule representing the action of function f on d using `clause` (fig. 3). The first argument to `clause` encodes variables corresponding to the above type, and it also includes the abstract representation of g on which the pattern matching should occur. The second argument to `clause`, which is of type `Term`, refers to the variables in the first argument using de-bruijn indices, and it encodes the output of the action of function f on d . The constructor `var` in `Pattern` is used to introduce new variables in the `clause` definition. The type `Pattern` also has another constructor `con` used to represent the pattern matching term g . The type `Term` has similar constructors `var` and `con`, but with different types, used to encode the output of the recursion rule. Given $\alpha : A \rightarrow W$, $\beta : B$ and $\omega : W$, the computation rule corresponding to the above type is given as follows.

$$f(P, d, g(\alpha, \beta, \omega)) \equiv d(\alpha, f \circ \alpha, \beta, \omega, f(\omega))$$

The second argument ($f \circ \alpha$) to d is the composite of function f and α . We can build a composite function inside a lambda using the constructor `lam`. The arguments to `lam` are referenced using de-bruijn indices inside the lambda body. So, the de-bruijn indices for referring variables outside the lambda body are updated accordingly. The clause definition, which evaluates to the above computation rule of W , is given in figure 3. The de-bruijn index reference increments right to left starting from the last argument ω . Inside the lambda body, the reference `0` refers to the lambda argument a , and the index references to the variables outside the lambda body starts from 1 and increments towards the left. The above

clause definition is defined using `defineFun` primitive, and the function f is brought into scope by `unquoteDecl`.

In the automation tool, `generateRec` interface is used to generate the recursion rule f for the type W . The implementation of `generateRec` is given as follows.

```

1 generateRec : Arg Name → Name → List Nat → TC T
2 generateRec (arg i f) t indLs =
3   bindTC (getConstructors t) λ cns →
4   bindTC (getLength cns) λ lcons →
5   bindTC (getClause lcons zero t f indLs cns) λ cls →
6   bindTC (getType t) λ RTy →
7   bindTC (getRtype t indLs zero RTy) λ funType →
8   bindTC (declareDef (arg i f) funType) λ _ →
9   (defineFun f cls)
10
11

```

`generateRec` uses `getClause` and `getRtype` to build the computation and elimination rules respectively. It takes three arguments which consists of the function name to be defined as an element of type `Arg Name`, the quoted `Name` of the type W and a list containing the index count of the individual constructors. `generateRec` can be used to automate the generation of recursion rules for the inductive types with point constructors. It can be used with the constructors of inductive types that takes no arguments (eg. `Bool`), one or multiple arguments (eg. `coproduct`, `cartesian product`) and the constructors that takes argument from the inductive type being defined (eg. `Nat`, `List`, `Vec`). The recursion rule generated by `generateRec` is brought into scope using `unquoteDecl` as follows.

```
unquoteDecl f = generateRec (vArg f) (quote W) (0 :: [])
```

The third argument to `generateRec` is a list consisting of the index count for the constructors. Since the constructor g does not have any index, the count is given as 0. It is required to pass the index count for each constructor explicitly as the Agda reflection library does not have built-in primitives to retrieve the index value.

3.2 Dependent Eliminator

The dependent eliminator or the induction principle is used to define the behavior of a mapping f' when the output type is dependent on the input element to f' . For the inductive type W , the induction principle says that to define a mapping $f' : (w : W) \rightarrow P(w)$, it suffices to define the action of f' on g . To define the action of f' on g , we need an element of the following type.

$$d' : \prod_{\alpha : A \rightarrow W} \left(\prod_{a : A} P(\alpha(a)) \right) \rightarrow \prod_{\beta : B} \prod_{\omega : W} P(\omega) \rightarrow P(g(\alpha, \beta, \omega))$$

In the first argument of d' , the type P depends on the action of the function α on input $a : A$. In the second argument, it depends on the input element $\omega : W$. The output of function d' depends on the action of constructor g on inputs α , β , and ω .


```

(pi (vArg
  (pi (vArg `A) (abs "-" `W)))
  (abs "-" (pi (vArg
    (pi (vArg `A)
      (abs "-" (var 2
        (vArg (var 1
          (vArg (var 0 [] :: []) :: []))))))
      (abs "-" (pi (vArg `B)
        (abs "-" (pi (vArg `W)
          (abs "-" (pi (vArg (var 4
            (vArg (var 0 [] :: []) :: []))
              (abs "-" (var 5
                (vArg (con (quote g)
                  (vArg (var 4 [] ::
                    (vArg (var 2 [] ::
                      (vArg (var 1 [] :: []) :: []))))))))))))))
      -- α : A → W
      -- A → P(α(a))
      -- a : A
      -- P(α(a))
      -- α ref
      -- a ref
      -- β : B
      -- ω : W
      -- P(ω)
      -- ω ref
      -- P(g i β ω)
      -- g ref
      -- α ref
      -- β ref
      -- ω ref
    )
  )
)

```

Figure 4. Abstract syntax tree for constructor d'

To build the dependent eliminator, we need the type of the function d' . We can construct the abstract syntax tree of d' using the static type information obtained from g . To construct d' , during the traversal of the abstract syntax tree of g , when we identify a function α with codomain W , we add a new function with the same argument A as in α and codomain P , which depends on the action of α on $(a : A)$. We copy the constant B directly without any changes as in the case of the non-dependant eliminator. When we see $(\omega : W)$ in a non-codomain position, we add the type P , which depends on the element ω , next to it. Finally, the output type W of the constructor g is changed to P , which depends on the action of the constructor g on inputs α, β and ω (fig. 4).

We can construct the type of the induction principle f' using d' . The type P in the mapping f' depends on the element of the input type W . The following equation gives the type of the induction principle f' .

$$\begin{aligned}
 f' : (P : W \rightarrow \text{Set}) \rightarrow \\
 (d' : \prod_{\alpha:A \rightarrow W} (\prod_{a:A} P(\alpha(a))) \rightarrow \prod_{\beta:B} \prod_{\omega:W} P(\omega) \rightarrow \\
 P(g(\alpha, \beta, \omega))) \rightarrow \\
 (w : W) \rightarrow P(w)
 \end{aligned}$$

The computation rule corresponding to the above type is the same as the computation rule of the recursion principle of W except that the function f and d are changed to f' and d' respectively. It is constructed using clause definitions following the same approach as the recursion principle. The computation rule corresponding to f' is given as follows.

$$f'(P, d', g(\alpha, \beta, \omega)) \equiv d'(\alpha, f' \circ \alpha, \beta, \omega, f'(\omega))$$

We can automate the generation of the induction rule f' and its corresponding computation rules for the type W using `generateInd` interface. The following code gives the implementation of `generateInd`.

```

1
2 generateInd : Arg Name → Name → List Nat → TC T
3 generateInd (arg i f) t indLs =
4   bindTC (getIndex t indLs) λ indLs' →
5   bindTC (getConstructors t) λ cons →
6   bindTC (getLength cons) λ lcons →
7   bindTC (getClauseDep lcons zero t f indLs' cons) λ cls →
8   bindTC (getType t) λ RTy →
9   bindTC (getRTyInd t zero indLs' RTy) λ funType →
10  bindTC (declareDef (arg i f) funType) λ _ →
11  (defineFun f cls)
12

```

`generateInd` uses `getClauseDep` to generate the clause definitions representing the computation rule of the type W . The abstract representation of the type of W is provided by `getRTyInd`. f' generated by `generateInd` is brought into scope by `unquoteDecl` as follows.

```
unquoteDecl f' = generateInd (vArg f') (quote W) []
```

We pass an empty list to `generateInd` as the type W has no index. We can also pass an empty list if all the constructors of a type has the same number of index as the parent type. But if any one constructor has an index count different from the index count of the parent type, then we have to explicitly pass the index count of all the constructors.

4 Code Generation for Higher Inductive Type

In Agda, there are no built-in primitives to support the definition of higher inductive types. However, we can still define a higher inductive type with a base type using Dan Licata's [4] method as discussed in section 2.1. In this section, we discuss the automation of code generation for the boiler-plate code segments defining the higher inductive type. We also describe how to automate the code generation for the elimination and the computation rules of the higher inductive type using static type information obtained from the base type.

4.1 Higher Inductive Type Definition

In Agda, we define an inductive type using `data` keyword. The data literal characterize a data type by declaring its type and specifying its constructors. Consider the following generic form for the definition of an inductive type W with constructors $g_1 \dots g_n$.

```

data W* (x1 : P1) ... (xn : Pn) : Q1 → ... → Qn → Set ℓ where
  g1* : {i1 : Q1} ... {in : Qn} → Type1 → W* x1 ... xn i1 ... in
  ...
  gn* : {j1 : Q1} ... {jn : Qn} → Typen → W* x1 ... xn j1 ... jn

```

The parameters $(x_1 : P_1) \dots (x_n : P_n)$ are common to all the constructors, and the type of each constructor implicitly encodes the parameter references. However, the indices

are different for each constructor. So, the type of each constructor explicitly exhibits the index declaration. During the construction of a higher inductive type, we have to represent the parameters and the indices as explicit arguments in the type of the constructors. We define a higher inductive type W as a top-level definition using a base type W^* similar to the module `Circle` in section 2.1. The reflection type Definition provides us the type and the constructors of the base type W^* . We copy the type of W^* to W and for the constructors $g_1 \dots g_n$ of W , we traverse the abstract representation of the type of $g_1^* \dots g_n^*$ respectively replacing the occurrence of W^* to W in every strict positive position. Consider a constructor g_i^* having the following type.

$$g_i^* : (A \rightarrow W^*) \rightarrow (B \rightarrow W^*) \rightarrow C \rightarrow W^* \rightarrow W^*$$

We built the type of g_i by traversing the abstract syntax tree of g_i^* and replacing the base type W^* with the higher inductive type W . The abstract syntax tree of g_i^* incorporates the type of the parameters and the indices if present. We have to retain the parameters and the indices explicitly during the construction of g_i . The following equation represents the type of the constructor g_i .

$$g_i : (A \rightarrow W) \rightarrow (B \rightarrow W) \rightarrow C \rightarrow W \rightarrow W$$

We explicitly pass the type of the path constructors to the automation tool. The higher inductive type definition of `Circle` in section 2.1 represents the path constructors as propositional equalities. The automation tool takes the path types as input and declares them as propositional equalities using the reflection primitive `declarePostulate`. We introduce a new data type `ArgPath` to input the path types to the automation tool.

$$\begin{aligned} \text{data ArgPath } \{\ell_1\} : \text{Set } (\text{Isuc } \ell_1) \text{ where} \\ \text{argPath} : \text{Set } \ell_1 \rightarrow \text{ArgPath} \end{aligned}$$

The constructor `argPath` takes the type of a path constructor as input. We define the generic form of a higher inductive type as follows.

$$\begin{aligned} \text{data-hit (quote } W^*) \ W \\ \text{Wpoints } (g_1 :: \dots :: g_n :: []) \\ \text{Wpaths } (p_1 :: \dots :: p_n :: []) \\ \text{(argPath} \\ \quad (\{x_1 : P_1\} \rightarrow \dots \rightarrow \{x_n : P_n\} \rightarrow \\ \quad \{i_1 : Q_1\} \rightarrow \dots \rightarrow \{i_n : Q_n\} \rightarrow \text{Type}_1 \rightarrow \\ \quad (g_i \{x_1\} \dots \{x_n\} \{i_1\} \dots \{i_n\} \dots) \equiv (g_j \dots)) :: \\ \quad \vdots \\ \text{argPath} \\ \quad (\{x_1 : P_1\} \rightarrow \dots \rightarrow \{x_n : P_n\} \rightarrow \\ \quad \{j_1 : Q_1\} \rightarrow \dots \rightarrow \{j_n : Q_n\} \rightarrow \text{Type}_n \rightarrow \\ \quad (g_i \{x_1\} \dots \{x_n\} \{j_1\} \dots \{j_n\} \dots) \equiv (g_j \dots)) :: []) \end{aligned}$$

We define holders `Wpoints` for point constructors and `Wpaths` for path constructors as part of the higher inductive type definition of W . We cannot retrieve the constructors of

the higher inductive type W using Definition. Therefore, `Wpoints` and `Wpaths` act as the only references for the constructors of W . The elements of the `argPath` list represent the type of the path constructors $p_1 \dots p_n$ respectively. We explicitly include the parameter references $\{x_1 : P_1\} \dots \{x_n : P_n\}$ and the index references $\{i_1 : Q_1\} \dots \{i_n : Q_n\}$ in the type of the arguments to `argPath`. The points $g_1 \dots g_n$ are not in scope when used in the identity type passed to `argPath`. The automation tool uses the base type constructors $g_1^* \dots g_n^*$ as dummy arguments in the place of $g_1 \dots g_n$ respectively. The automation tool implements the interface `data-hit` as follows.

```

1
2 data-hit : ∀{ℓ₁} (baseType : Name) → (indType : Name) →
3   (pointHolder : Name) → (lcons : List Name) →
4   (pathHolder : Name) → (lpaths : List Name) →
5   (lpathTypes : (List (ArgPath {ℓ₁}))) → TC ⊤
6 data-hit base ind pointH lcons pathH lpaths lpathTypes =
7   bindTC (defineHindType base ind) λ _ →
8   bindTC (getConstructors base) λ lcons' →
9   bindTC (defineHitCons base ind lcons' lcons) λ _ →
10  bindTC (getPathTypes base ind lcons' lcons lpathTypes) λ lp →
11  bindTC (defineHitPathCons lpaths lp) λ _ →
12  bindTC (definePointHolder pointHolder lcons) λ _ →
13  definePathHolder pathHolder lpaths
14

```


The higher inductive type W , the points $g_1 \dots g_n$, the paths $p_1 \dots p_n$, and the holders `Wpoints` and `Wpaths` are brought into scope by `unquoteDecl`. In the above implementation of `data-hit`, `defineHindType` defines the higher inductive type as a top-level definition using the base type. `defineHitCons` specifies the point constructors of the higher inductive type using the type information obtained from the constructors of the base type, and `defineHitPathCons` builds the paths constructors of the higher inductive type using the `argPath` list. The following code automates the generation of the higher inductive type definition for `Circle` given in section 2.1.

```

unquoteDecl S Spoints base Spaths loop =
  data-hit (quote S*) S
    Spoints (base :: []) -- point constructors
    Spaths (loop :: []) -- path constructors
    (argPath (base* ≡ base*) :: []) -- base replaces base*

```

The identity type input (`base* ≡ base*`) to `argPath` represents the type of the path loop, and it uses the inductive type constructor `base*` as a dummy argument in the place of the higher inductive type constructor `base`. The constructor `base` comes into scope only during the execution of `unquoteDecl`, and so cannot be used in the identity type reference in `argPath`. We use the constructor `base*` of type `S*` as dummy argument because the type of `base*` is similar to `base`, and has the same references for the common arguments. The automation tool traverses the abstract syntax tree of `loop` and replaces the occurrences of `base*` with `base`.

Figure 5. Abstract syntax tree for constructor f

4.2 Non-dependent Eliminator

Non-dependent eliminator or the recursion principle of a higher inductive type W maps the points and paths of W to an output type P . Let $g_1 : A \rightarrow W$ and $g_2 : B \rightarrow W$ be the point constructors, and $l : (\alpha : A) \rightarrow (\beta : B) \rightarrow (g_1 \alpha) \equiv (g_2 \beta)$ be the path constructor of W . To define a mapping $f : W \rightarrow P$, we need points $d_1 : A \rightarrow P$ and $d_2 : B \rightarrow P$ and a path $h : (\alpha : A) \rightarrow (\beta : B) \rightarrow (d_1 \alpha) \equiv (d_2 \beta)$ in the space P . The path constructor l can take arguments of any type including the higher inductive type W but for simplicity we will consider only arguments of constant type.

To construct the recursion principle f , we need to build the type of points d_1, d_2 and path h . The type of d_1 and d_2 are built from the abstract syntax tree of points g_1 and g_2 respectively using the approach described in section (3.1). We build the type of h by traversing the abstract syntax tree of l . During the traversal, we replace the points g_1 and g_2 , which are arguments to the identity type, in the co-domain of the path l by the points d_1 and d_2 respectively. In figure (5), the arguments of the identity type in the abstract syntax tree of h references the points d_1 and d_2 using de-bruijn index 3 and 2 respectively. We copy the other two arguments of l , which are of constant type A and B , without any changes into the type of h .

The recursion rule f corresponding to figure (5) is given by the following equation.

$$f : (P : \text{Set}) \rightarrow (d_1 : A \rightarrow P) \rightarrow (d_2 : B \rightarrow P) \rightarrow (h : (\alpha : A) \rightarrow (\beta : B) \rightarrow (d_1 \alpha) \equiv (d_2 \beta)) \rightarrow W \rightarrow P$$

The computation rules for constructors g_1 and g_2 give the action of eliminator f on input $g_1(\alpha)$ and $g_2(\beta)$ respectively.

The computation rules say that the function f maps the points $g_1(\alpha)$ and $g_2(\beta)$ to the points $d_1(\alpha)$ and $d_2(\beta)$ respectively in the output type P . We specify the clause definition to build the computation rules for g_1 and g_2 in the same way as for a regular inductive type (sec. 3.1) except that the clause definition for the higher inductive type W has an additional reference variable to the path h . The following definitional equalities give the computation rules for g_1 and g_2 .

$$f(P, d_1, d_2, h, g_1(\alpha)) \equiv d_1(\alpha) \\ f(P, d_1, d_2, h, g_2(\beta)) \equiv d_2(\beta)$$

The computation rule for the path constructor l defines the action of function f on the path $(l \alpha \beta)$. In type theory, the functions are functorial, and they preserve the path structures in their mapping to the output type [7]. The function $ap_f(1)$ defines the action of the non-dependent eliminator f on the path l . Using (1), we can build the computation rule for the path l as follows.

$$\beta_f : (P : \text{Set}) \rightarrow (d_1 : A \rightarrow P) \rightarrow (d_2 : B \rightarrow P) \rightarrow (h : (\alpha : A) \rightarrow (\beta : B) \rightarrow (d_1 \alpha) \equiv (d_2 \beta)) \rightarrow \{\alpha : A\} \rightarrow \{\beta : B\} \rightarrow ap_{f(\lambda x \rightarrow f(P, d_1, d_2, h, x))}(l \alpha \beta) \equiv (h \alpha \beta)$$

The computation rule β_f exists only as propositional equality. We build the type of β_f using the same approach as for the recursion rule f . The type of f and β_f is identical except for the mapping $W \rightarrow P$ in f which is replaced by the term representing the action of function f on the path $(l \alpha \beta)$. The function ap applies f , which is nested inside a lambda function, on the arguments $(g_1 \alpha)$ and $(g_2 \beta)$ of the path $(l \alpha \beta)$. We use the constructor `lam` of `Term` to introduce the lambda function with argument x (fig. 6). Inside the lambda body, the argument x gets a de-bruijn index reference of zero and we update the remaining de-bruijn indices accordingly. The application of function f to the path $(l \alpha \beta)$ substitutes the points $(g_1 \alpha)$ and $(g_2 \beta)$ for the lambda argument x , and it evaluates to the path $(h \alpha \beta)$ in the output type P . We implement the `generateRecHit` interface as follows.

```

1 generateRecHit : Arg Name -> List (Arg Name) ->
2   (baseType : Name) -> (indexList : List Nat) ->
3   (baseRec : Name) -> (indType : Name) ->
4   (points : List Name) -> (paths : List Name) -> TC T
5 generateRecHit (arg i f) argD base il rec ind points paths =
6   bindTC (getConstructors base) \ lcons ->
7   bindTC (getLength points) \ lpoints ->
8   bindTC (getLength paths) \ lpaths ->
9   bindTC (getPathClause lpoints lpaths rec) \ clause ->
10  bindTC (getType base) \ RTy ->
11  bindTC (getRTyPath base ind rec il paths zero RTy) \ fty ->
12  bindTC (declareDef (arg i f) fty) \ _ ->
13  bindTC (defineFun f clause) \ _ ->
14  (generateβRecHit argD base il rec ind f points paths)
15
16
```

`generateRecHit` takes the base type recursion rule as input and uses that to map the points $g_1(\alpha)$ and $g_2(\beta)$ to




Figure 6. Abstract syntax tree for the term representing the action of function f on path l

$d_1(\alpha)$ and $d_2(\beta)$ respectively in the abstract syntax tree of the path $(l \alpha \beta)$. The second argument `argD` is a list of terms representing the computation rules for the path constructors. The `generateβRecHit` interface takes `argD` as input and builds the computation rule for the path constructor l . Other inputs to `generateRecHit` are the point and path holders declared during the higher inductive type definition of W . The following automation code generates the recursion rule and the computation rules for `Circle`.

```

unquoteDecl recS1* = generateRec (vArg recS1*)
  (quote S1*) (0 :: [])

unquoteDecl recS1 βrecS1 = generateRecHit (vArg recS1)
  ((vArg βrecS1) :: [])
  (quote S1*) (0 :: [])
  (quote recS1*)
  (quote S1) S1points S1paths

```

The term βrecS_1 represents the computation rule for the path constructor `loop` and is brought into scope by `unquoteDecl`.

References

- [1] 2017. *Agda's Documentation*. <http://agda.readthedocs.io/en/latest/language/reflection.html>.
- [2] Carlo Angiuli, Edward Morehouse, Daniel R. Licata, and Robert Harper. 2014. Homotopical Patch Theory. In *Proceedings of the 19th ACM SIGPLAN International Conference on Functional Programming (ICFP '14)*. ACM.
- [3] David Christiansen and Edwin Brady. 2016. Elaborator Reflection: Extending Idris in Idris. In *Proceedings of the 21st ACM SIGPLAN International Conference on Functional Programming (ICFP '16)*. ACM, Nara, Japan.
- [4] Daniel R. Licata. 2011. Running Circles Around (In) Your Proof Assistant; or, Quotients that Compute. (April 2011). <https://homotopytypetheory.org/2011/04/23/running-circles-around-in-your-proof-assistant>.
- [5] Ulf Norell. 2007. Towards a practical programming language based on dependent type theory. (2007).
- [6] Coq Development Team. 2009. The Coq proof assistant reference manual, version 8.2. (2009). <https://coq.inria.fr/>.
- [7] The Univalent Foundations Program. 2013. *Homotopy Type Theory: Univalent Foundations of Mathematics*. <https://homotopytypetheory.org/book>, Institute for Advanced Study.