

# Encoding Higher Inductive Types Without Boilerplate

## A Study in Agda Metaprogramming

Paventhan Vivekanandan  
School of Informatics, Computing and Engineering  
Indiana University  
Bloomington, Indiana, USA  
pvivekan@indiana.edu

David Thrane Christiansen  
Galois, Inc.  
Portland, Oregon, USA  
dtt@galois.com

### Abstract

Higher inductive types are inductive types that include non-trivial higher-dimensional structure, represented as identifications at the type that are not reflexivity. While work proceeds on type theories with a computational interpretation of univalence and higher inductive types, it is convenient to encode these structures in more traditional type theories. However, these encodings involve a great deal of error-prone additional syntax. We present a library that uses Agda’s metaprogramming facilities to automate this process, allowing higher inductive types to be specified with minimal additional syntax.

**Keywords** Higher inductive type, Elaboration, Elimination rules, Computation rules

### 1 Introduction

A metaprogram is a program used to create other programs. In dependently-typed programming languages like Agda and Coq, a reflection library provides metaprogramming interfaces to support code generation for proofs and programs. Elaborator reflection [8], a new paradigm for metaprogramming, provides the language users with a powerful set of interfaces for efficient code generation. An *elaborator* is a metaprogram that converts a high-level language syntax in its abstract representation to a core language which is then type checked independently. By exposing a primitive monad to elaboration framework in Idris, Christiansen [8] showed how to write termination proofs for general recursive functions by automating Bove-Capretta [5] transformation, and how to generate data types matching schemas of external data sources based on Idris’s type providers [6].

Agda is a proof-assistant based on Martin-Löf’s intensional type theory. Unlike Idris, which has a dedicated elaborator to convert the high-level syntax to its core language, Agda’s elaboration mechanism is a part of its type checker. Inspired by Idris’s elaborator reflection, Agda development team extended the reflection library of Agda by exposing its elaboration monad to its high-level metaprogramming instructions. The elaboration monad provides an interface

to the Agda type checker through a set of primitive operations which can be used to retrieve static type information regarding various code segments. The primitive operations can also be used to build code fragments using constructs of an abstract syntax tree, and to convert an abstract syntax tree to its concrete syntax.

In this paper, we discuss metaprogramming using elaborator reflection in Agda. By using the elaboration monad, we performed code generation for the elimination and computation rules of inductive types and higher inductive types. We also automated the code generation for the boiler-plate code segment used to define the higher inductive type. We discuss the construction process in detail and introduce new metaprogramming interfaces which build the code segments using elaborator reflection. The interfaces extensively use Agda’s reflection primitives and build the abstract syntax tree of the code segments with static type information obtained using the reflection primitives. The generated code is then brought into scope by another top-level reflection primitive.

Elaborator reflection is a powerful mechanism which allows us to create more interesting metaprograms. By performing the above automation, we abstracted the difficulties involved in implementing and using inductive types and higher inductive types by automating the construction of their corresponding elimination and computation rules. We also achieved an extensive reduction in code size, and we demonstrated how to extend a language without modifying the compiler or depending on the language implementers. More specifically, we discuss the following contributions in this paper.

- We automate the construction of the recursion and the induction principles for inductive types with constructors taking zero arguments, one or more arguments, and the type being defined itself as an argument.
- We discuss the automation of code generation of boiler-plate code for a higher inductive type defined inside a module. The boiler-plate code depends on a base type defined as private inside the module. The constructors of the private base type are not accessible outside the module.

- We discuss the automation of code generation of the elimination and the computation rules for a higher inductive type with point and path constructors. We demonstrate the code generation of the reduction rules, specified as postulates, for the path constructors of the higher inductive type.
- We discuss the automation of code generation for patch theory [4] implementation enriched with patches of encryption. We also examine the usage of the automation tool in a cryptography application [18].

The automation tool abstracts the implementation difficulties of a higher inductive type and its recursion and induction principles. We automated the code generation of approximately 1500 lines of code of the patch theory implementation with just 70 lines of automation code. The automation code extensively uses the static type information queried from the type checker using the reflection primitives.

## 2 Background

### 2.1 Agda Reflection

Agda's reflection library enables compile-time metaprogramming. This reflection library directly exposes parts of the implementation of Agda's type checker and elaborator for use by metaprograms, in a manner that is similar to Idris's elaborator reflection [7, 8] and Lean's tactic metaprogramming [10]. The type checker's implementation is exposed as effects in a monad called TC.

Agda exposes a representation of its syntax to metaprograms, including datatypes for expressions (called *Term*) and definitions (called *Definition*). The primitives exposed in TC include declaring new metavariables, unifying two *Terms*, declaring new definitions, adding new postulates, computing the normal form or weak head normal form of a *Term*, inspecting the current context, and constructing fresh names. This section describes the primitives that are used in our code generation library; more information on the reflection library can be found in the Agda documentation [1].

TC computations can be invoked in three ways: by macros, which work in expression positions, using the `unquoteDecl` operator in a declaration position, which can bring new names into scope, and using the `unquoteDef` operator in a declaration position, which can automate constructions using names that are already in scope. This preserves the principle in Agda's design that the system never invents a name.

An Agda *macro* is a function of type  $t_1 \rightarrow t_2 \rightarrow \dots \rightarrow \text{Term} \rightarrow \text{TC } \top$  that is defined inside a `macro` block. Macros are special: their last argument is automatically supplied by the type checker, and consists of a *Term* that represents the metavariable to be solved by the macro. If the remaining arguments are quoted names or *Terms*, then the type checker will automatically quote the arguments at the macro's use

```
macro
  mc1 : Term → Term → TC ⊤
  mc1 exp hole =
    do exp' ← quoteTC exp
    unify hole exp'

sampleTerm : Term
sampleTerm = mc1 (λ (n : Nat) → n)
```

**Figure 1.** A macro that quotes its argument

```
macro
  mc2 : Term → Term → TC ⊤
  mc2 exp hole =
    do exp' ← unquoteTC exp
    unify hole exp'

sampleSyntax : Nat → Nat
sampleSyntax =
  mc2 (lam visible (abs "n" (var 0 [])))
```

**Figure 2.** A macro that unquotes its argument

site. At some point, the macro is expected to unify the provided metavariable with some other term, thus solving it.

Figure 1 demonstrates a macro that quotes its argument. The first step is to quote the quoted expression argument again, using `quoteTC`, yielding a quotation of a quotation. The result of this double-quotation is passed, using Agda's new support for Haskell-style `do`-notation, into a function that unifies it with the hole. Because unification removes one layer of quotation, `unify` inserts the original quoted term into the hole. The value of `sampleTerm` is

```
lam visible (abs "n" (var 0 []))
```

The constructor `lam` represents a lambda, and its body is formed by the abstraction constructor `abs` that represents a scope in which a new name "n" is bound. The body of the abstraction is a reference back to the abstracted name using de Bruijn index 0.

The `unquoteTC` primitive removes one level of quotation. Figure 2 demonstrates the use of `unquoteTC`. The macro `mc2` expects a quotation of a quotation, and substitutes its unquotation for the current metavariable.

The `unquoteDecl` and `unquoteDef` primitives, which run TC computations in a declaration context, will typically introduce new declarations by side effect. A function of a given type is declared using `declareDef`, and it can be given a definition using `defineFun`. Similarly, a postulate of a given type is defined using `declarePostulate`. Figure 3 shows an Agda implementation of addition on natural numbers, while figure 4 demonstrates an equivalent metaprogram that adds the same definition to the context.

In Figure 4, `declareDef` declares the type of `plus`. The constructor `pi` represents dependent function types, but

```

plus : Nat → Nat → Nat
plus zero b = b
plus (suc n) b = suc (plus n b)

```

**Figure 3.** Addition on natural numbers

```

pattern vArg x = arg (arg-info visible relevant) x
pattern _`⇒_ a b = pi (vArg a) (abs "_" b)
pattern `Nat = def (quote Nat) []

unquoteDecl plus =
  do declareDef (vArg plus) (`Nat `⇒ `Nat `⇒ `Nat)
  defineFun plus
    (clause (vArg (con (quote zero) [])) ::
      vArg (var "y") ::
      [])
    (var 0 []) ::
    clause (vArg (con (quote suc)
      (vArg (var "x") :: [])) ::
      vArg (var "y") ::
      []))
    (con (quote suc)
      (vArg (def plus
        (vArg (var 1 []) ::
          vArg (var 0 []) :: [])) ::
        [])) ::
    []

```

**Figure 4.** Addition, defined by metaprogramming

a pattern synonym is used to make it shorter. Similarly, `def` constructs references to defined names, and the pattern synonym ``Nat` abbreviates references to the defined name `Nat`, and `vArg` represents the desired visibility and relevance settings of the arguments. Once declared, `plus` is defined using `defineFun`, which takes a name and a list of clauses, defining the function by side effect. Each clause consists of a pattern and a right-hand side. Patterns have their own datatype, while right-hand sides are `Terms`. The name `con` is overloaded: in patterns, it denotes a pattern that match a particular constructor, while in `Terms`, it denotes a reference to a constructor.

## 2.2 Higher Inductive Types

Homotopy type theory [17] is a research program that aims to develop univalent, higher-dimensional type theories. A type theory is *univalent* when equivalences between types are considered equivalent to equalities between types; it is *higher-dimensional* when we allow non-trivial identifications that every structure in the theory must nevertheless respect. Identifications between elements of a type are considered to be at the lowest dimension, while identifications between identifications at dimension  $n$  are at dimension  $n + 1$ . Voevodsky added univalence to type theories as an axiom, asserting new identifications without providing a means to compute with them. While more recent work arranges the computational mechanisms of the type theory such that univalence

can be derived, as is done in cubical type theories, we are concerned with modeling concepts from homotopy type theory in existing, mature implementations of type theory, so we follow Univalent Foundations Program [17] in modeling paths using Martin-Löf's identity type. Higher-dimensional structure can arise from univalence, but it can also be introduced by defining new type formers that introduce not only introduction and elimination principles, but also new non-trivial identifications.

In homotopy type theories, one tends to think of types not as collections of distinct elements, but rather through the metaphor of topological spaces. The individual elements of the type correspond with points in the topological space, and identifications correspond to paths in this space.

While there is not yet a general schematic characterization of a broad class of higher inductive types along the lines of Dybjer's inductive families, it is convenient to syntactically represent the higher inductive types that we know are acceptable as if we had such a syntax. Thus, we sometimes specify a higher inductive type similarly to a traditional inductive type by providing its constructors (*i.e.* its points); we additionally specify the higher-dimensional structure by providing additional constructors for paths. For example, figure 5 describes `Circle`, which is a higher inductive type with one point constructor `base` and one non-trivial path constructor `loop`.

```

data Circle : Set where
  base : Circle
  loop : base ≡ base

```

**Figure 5.** A specification of a higher inductive type

Agda is a programming language that was originally an implementation of Luo's UTT extended with primitive dependent pattern matching, itself a derivative of the Calculus of Constructions and Martin-Löf's intensional type theory. Agda's type theory has since gained a number of new features, among them the ability to restrict pattern matching to that subset that does not imply Streicher's Axiom K, which is inconsistent with univalence. The convenience of programming in Agda, combined with the ability to avoid axiom K, makes it a good laboratory for experimenting with the idioms and techniques of univalent programming while more practical implementations of univalent type theories are under development.

In Agda, we don't have built-in primitives to support the definition of higher inductive type such as `Circle`. One approach is to use Agda's rewrite rules mechanism to define higher inductive types. In this approach, we define the dependent and non-dependent eliminators of a higher inductive type as parameterized modules inside which we declare the computation rules for points as rewrite rules

cite Coqu  
group, M  
and Harp

cite 1994

Cite Ulf's  
official w

cite Luo's

Cite Cock

cite Cock  
Abel, exte  
abstract a  
2016

using `{-# REWRITE , ... #-}` pragma. However, Agda's reflection library do not have interfaces to support introducing new pragmas and defining new modules. Another approach to define higher inductive types is to use Dan Licata's method [11]. According to this method, a higher inductive type is defined using type abstraction inside a module. The module consists of a boiler-plate code segment which defines the higher inductive type using a private base type. Inside the module, the recursion and the induction principles acts on the constructors of the private base type. The abstract type is then exported allowing the reduction rules for point constructors to hold definitionally. For example, `Circle` is defined using Dan Licata's method as follows.

```

1
2 module Circle where
3   private
4     data S* : Set where
5       base* : S*
6
7     S : Set
8     S = S*
9
10    base : S
11    base = base*
12
13    postulate
14      loop : base ≡ base*
15
```

Inside the module `Circle`, the type `S` is defined using a private type `S*`. The constructor `base` is defined using `base*` and the path `loop` is given as a propositional equality. The recursion and the induction principles are defined to pattern match on the constructor `base*` of the type `S*`. The clients of `Circle` will not have access to the constructor `base*` of the private type `S*` as it is not visible outside the module. The clients only way of access to the constructor is through the non-dependent and dependent elimination rules. The following code gives the non-dependent eliminator or the recursion rule `recS`.

```

1
2 recS : {C : Set} →
3   (cbase : C) →
4   (cloop : cbase ≡ cbase) →
5   S → C
6 recS cbase cloop base* = cbase
7
8 postulate
9   βrecS : {C : Set} →
10    (cbase : C) →
11    (cloop : cbase ≡ cbase) →
12    ap (recS cbase cloop) loop ≡ cloop
13
```

`recS` ignores the path argument and simply computes to the appropriate answer for the point constructor. The computational behavior for the path constructor `loop` is postulated using reduction rule  $\beta_{\text{recS}}$ . In the above code, the function `ap` gives the action of a function  $f$  on paths. The following

equation gives the type of `ap`.

$$ap_f : (x =_A y) \rightarrow (f(x) =_A f(y)) \quad (1)$$

The following code gives the dependent eliminator or the induction rule `indS` and its computational rules.

```

1
2 indS : {C : S → Set} →
3   (cbase : C base) →
4   (cloop : transport C loop cbase ≡ cbase) →
5   (circle : S) → C circle
6 indS cbase cloop base* = cbase
7
8 postulate
9   βindS : {C : S → Set} →
10    (cbase : C base) →
11    (cloop : transport C loop cbase ≡ cbase) →
12    apd (indS {C} cbase cloop) loop ≡ cloop
13
```

In the above code, the function `apd` gives the action of dependent functions of type  $f : \prod_{(x:A)} B(x)$  on paths. The following equation gives the type of `apd`.

$$apd_f : \prod_{p:x=y} (p_*(f(x)) =_{B(y)} f(y)) \quad (2)$$

In (2), the point  $p_*(f(x))$  lying in space  $B(y)$  can be thought of as an endpoint of a path obtained by lifting the path  $p$  from  $f(x)$  to a path in the total space  $\sum_{(x:A)} B(x) \rightarrow A$ . The following equation gives the type of  $p_*$  which is also known as *transport* [17].

$$transport_p^B : B(x) \rightarrow B(y) \quad (3)$$

where  $p : x = y$  for  $x, y : A$ .

In the next section, we will see automation of code generation for normal inductive types. In section 4, we will see automation of code generation for higher inductive types, and in section 5, we will revisit patch theory [4] application and discuss automation of code generation for the higher inductive types defined to model Darc's version control system.

### 3 Code Generation for Inductive Type

An inductive type  $X$  is a type that is freely generated by a finite collection of constructors. The constructors of  $X$  are functions with zero or more arguments and codomain  $X$ . The constructors can also take an element of type  $X$  itself as an argument, but only strictly positively. In Agda reflection library, data-type of type `Definition` stores the constructors of an inductive type as a list of `Name`. The type of a constructor can be retrieved by giving its `Name` as an input to the `getType` primitive. In the following subsections, we will discuss how to use the constructor informations to generate code for the elimination rules of an inductive type.

#### 3.1 Non-dependent Eliminator

For an inductive type  $W$ , with a constructor  $g$ , the recursion principle says that to define a mapping  $f : W \rightarrow P$ , it

```

(pi (vArg
  (pi (vArg `A) (abs "-" `W))) -- f1 : A → W
  (abs "-" (pi (vArg `B) -- B
    (abs "-" (pi (vArg `W) -- W
      (abs "-" `W)))))) -- W

```

Figure 6. Abstract syntax tree for constructor  $g$ 

suffices to define the action of  $f$  on input  $g$ . For example, the recursion principle for  $\text{Nat}$  says that a mapping  $f : \text{Nat} \rightarrow P$  can be given by defining the action of  $f$  on the constructors zero and  $(\text{succ } x)$  for  $x : \text{Nat}$ .

The constructor  $g$  can take zero or more arguments including an element of the type  $W$ . Lets define the type of  $g$  as follows.

$$g : (A \rightarrow W) \rightarrow B \rightarrow W \rightarrow W$$

To define the action of  $f$  on input  $g$ , we need a function  $d$  of the following type.

$$d : (A \rightarrow W) \rightarrow (A \rightarrow P) \rightarrow B \rightarrow W \rightarrow P \rightarrow P$$

The function  $f$  will map the constructor  $g$  to  $d$ . To construct the type of the recursion rule for  $W$ , we need to build the type of  $d$ . We can retrieve the static type information of  $g$  using reflection primitives, and use that to construct the type of  $d$ . The constructor  $\text{pi}$  of type  $\text{Term}$  encodes the abstract syntax tree (AST) representation of  $g$  (fig. 6). We can retrieve and traverse the AST of  $g$ , and add new type information into it to build a new type representing  $d$ .

During the traversal of the abstract syntax tree of  $g$ , when we identify a function  $f_1$  with codomain  $W$ , we add a new function  $f_2$ , with the same arguments as  $f_1$  and codomain  $P$ , to the tree. For example, in figure 6, a new function is built from the first argument  $A \rightarrow W$  by modifying the codomain  $W$  to  $P$ . The new function of type  $A \rightarrow P$  is added after the function  $A \rightarrow W$  as in figure 7. Constant types require no modifications. Therefore, we copy  $B$  into the new type without any changes. When the type  $W$  occurs directly in a non-codomain position, we add the type  $P$  next to it. Finally, we change the codomain  $W$  of  $g$  to  $P$  resulting in an abstract syntax tree representation of  $d$  (fig. 7). We repeat this process for all the constructors of a given type. In the case of  $\text{Nat}$ , we construct the output types for zero and  $(\text{succ } x)$  for  $x : \text{Nat}$  using the above method.

The parameter and the index of  $W$ , if present, should be encoded as part of the type of function  $f$ . We can retrieve information about the parameter and the index of  $W$  from its type. The constructors refer to the parameter and the index using de Bruijn indices. During the construction of the output type  $d$ , we should update the de Bruijn indices accordingly. The constructor  $\text{data-type}$  contains the count  $cp$  of parameters occurring in a defined type. It also encodes the constructors of the type as a list of  $\text{Name}$ . We can retrieve

```

(pi (vArg
  (pi (vArg `A) (abs "-" `W))) -- f1 : A → W
  (abs "-" (pi (vArg
    (pi (vArg `A) (abs "-" `P))) -- f2 : A → P
    (abs "-" (pi (vArg `B) -- B
      (abs "-" (pi (vArg `W) -- W
        (abs "-" (pi (vArg `P) -- P
          (abs "-" `P)))))))))) -- P

```

Figure 7. Abstract syntax tree for constructor  $d$ 

the index count by finding the difference between  $cp$  and length of the constructor list. The parameters are common to all the constructors of a type. But the index values are different for each constructor. So we have to encode unique indices for each constructor. Also, some constructors might not take the same number of indices as the parent type. For example, in the case of  $\text{Vec}$ , the constructor  $[]$  excludes the index  $\text{Nat}$  from its type. We do not have any reflection primitive to retrieve the index count from a constructor name. A workaround is to pass the index count of each constructor explicitly to the automation tool.

Once we have the type of  $d$ , we can build the type of the recursion rule for  $W$ . To encode the mapping  $W \rightarrow P$  in the recursion type we need to declare  $P$ . We can use the constructor  $\text{agda-sort}$  to introduce the type  $(P : \text{Set})$ . The type of the recursion rule  $f$  is given as follows.

$$\begin{aligned}
 f : (P : \text{Set}) \rightarrow \\
 (d : (A \rightarrow W) \rightarrow (A \rightarrow P) \rightarrow B \rightarrow W \rightarrow P \rightarrow P) \rightarrow \\
 W \rightarrow P
 \end{aligned}$$

The above type is declared using `declareDef`. We can build the computation rule representing the action of function  $f$  on  $d$  using `clause` (fig. 8). The first argument to `clause` encodes variables corresponding to the above type, and it also includes the abstract representation of  $g$  on which the pattern matching should occur. The second argument to `clause`, which is of type  $\text{Term}$ , refers to the variables in the first argument using de Bruijn indices, and it encodes the output of the action of function  $f$  on  $d$ . The constructor `var` in `Pattern` is used to introduce new variables in the `clause` definition. The type `Pattern` also has another constructor `con` used to represent the pattern matching term  $g$ . The type `Term` has similar constructors `var` and `con`, but with different types, used to encode the output of the recursion rule. Given  $\alpha : A \rightarrow W$ ,  $\beta : B$  and  $\omega : W$ , the computation rule corresponding to the above type is given as follows.

$$f(P, d, g(\alpha, \beta, \omega)) \equiv d(\alpha, f \circ \alpha, \beta, \omega, f(\omega))$$

The second argument  $(f \circ \alpha)$  to  $d$  is the composite of function  $f$  and  $\alpha$ . We can build a composite function inside a lambda using the constructor `lam`. The arguments to `lam` are referenced using de Bruijn indices inside the lambda body.

```

(clause
  (vArg (var "P") ::                -- P
    vArg (var "d") ::                -- d
    vArg (con (quote g)
      (vArg (var "alpha") ::        -- α
        vArg (var "beta") ::        -- β
        vArg (var "omega") :: [])) :: [])) -- ω
  (var 3
    (vArg (var 2 []) ::              -- α
      vArg (lam visible (abs "a"
        (def f
          (vArg (var 3
            (vArg (var 0 []) :: []) :: []))) ::
          vArg (var 1 []) ::          -- β
          vArg (var 0 []) ::          -- ω
          vArg (def f (vArg (var 0 []) :: [])))))) -- f(ω)

```

**Figure 8.** Clause definition for the computation rule of  $W$

So, the de Bruijn indices for referring variables outside the lambda body are updated accordingly. The clause definition, which evaluates to the above computation rule of  $W$ , is given in figure 8. The de Bruijn index reference increments right to left starting from the last argument  $\omega$ . Inside the lambda body, the reference 0 refers to the lambda argument  $a$ , and the index references to the variables outside the lambda body starts from 1 and increments towards the left. The above clause definition is defined using `defineFun` primitive, and the function  $f$  is brought into scope by `unquoteDecl`.

In the automation tool, `generateRec` interface is used to generate the recursion rule  $f$  for the type  $W$ . The implementation of `generateRec` is given as follows.

```

generateRec : Arg Name → Name → (indexList : List Nat) → TC T
generateRec (arg i f) t indLs =
  do indLs' ← getIndex t indLs
  cns ← getConstructors t
  lcons ← getLength cns
  cls ← getClause lcons zero t f indLs cns
  RTy ← getType t
  funType ← getRType t indLs' zero RTy
  declareDef (arg i f) funType
  defineFun f cls

```

`generateRec` uses `getClause` and `getRType` to build the computation and elimination rules respectively. It takes three arguments which consists of the function name to be defined as an element of type `Arg Name`, the quoted Name of the type  $W$  and a list containing the index count of the individual constructors. `generateRec` can be used to automate the generation of recursion rules for the inductive types with point constructors. It can be used with the constructors of inductive types that takes no arguments (eg. `Bool`), one or multiple arguments (eg. `coproduct`, `cartesian product`) and the constructors that takes argument from the inductive type being defined (eg. `Nat`, `List`, `Vec`). The recursion rule generated by `generateRec` is brought into scope using `unquoteDecl` as follows.

```
unquoteDecl f = generateRec (vArg f) (quote W) (0 :: [])
```

The third argument to `generateRec` is a list consisting of the index count for the constructors. Since the constructor  $g$  does not have any index, the count is given as 0. It is required to pass the index count for each constructor explicitly as the Agda reflection library does not have built-in primitives to retrieve the index value.

### 3.2 Dependent Eliminator

The dependent eliminator or the induction principle is used to define the behavior of a mapping  $f'$  when the output type is dependent on the input element to  $f'$ . For the inductive type  $W$ , the induction principle says that to define a mapping  $f' : (w : W) \rightarrow P(w)$ , it suffices to define the action of  $f'$  on  $g$ . To define the action of  $f'$  on  $g$ , we need an element of the following type.

$$d' : \prod_{\alpha:A \rightarrow W} (\prod_{a:A} P(\alpha(a))) \rightarrow \prod_{\beta:B} \prod_{\omega:W} P(\omega) \rightarrow P(g(\alpha, \beta, \omega))$$

In the first argument of  $d'$ , the type  $P$  depends on the action of the function  $\alpha$  on input  $a : A$ . In the second argument, it depends on the input element  $\omega : W$ . The output of function  $d'$  depends on the action of constructor  $g$  on inputs  $\alpha$ ,  $\beta$ , and  $\omega$ .

To build the dependent eliminator, we need the type of the function  $d'$ . We can construct the abstract syntax tree of  $d'$  using the static type information obtained from  $g$ . To construct  $d'$ , during the traversal of the abstract syntax tree of  $g$ , when we identify a function  $\alpha$  with codomain  $W$ , we add a new function with the same argument  $A$  as in  $\alpha$  and codomain  $P$ , which depends on the action of  $\alpha$  on  $(a : A)$ . We copy the constant  $B$  directly without any changes as in the case of the non-dependant eliminator. When we see  $(\omega : W)$  in a non-codomain position, we add the type  $P$ , which depends on the element  $\omega$ , next to it. Finally, the output type  $W$  of the constructor  $g$  is changed to  $P$ , which depends on the action of the constructor  $g$  on inputs  $\alpha$ ,  $\beta$  and  $\omega$  (fig. 9).

We can construct the type of the induction principle  $f'$  using  $d'$ . The type  $P$  in the mapping  $f'$  depends on the element of the input type  $W$ . The following equation gives the type of the induction principle  $f'$ .

$$\begin{aligned}
 f' : (P : W \rightarrow Set) \rightarrow \\
 (d' : \prod_{\alpha:A \rightarrow W} (\prod_{a:A} P(\alpha(a))) \rightarrow \prod_{\beta:B} \prod_{\omega:W} P(\omega) \rightarrow \\
 P(g(\alpha, \beta, \omega))) \rightarrow \\
 (w : W) \rightarrow P(w)
 \end{aligned}$$

The computation rule corresponding to the above type is the same as the computation rule of the recursion principle of  $W$  except that the function  $f$  and  $d$  are changed to  $f'$  and  $d'$  respectively. It is constructed using clause definitions

```

(pi (vArg
  (pi (vArg `A) (abs "-" `W)))
  (abs "-" (pi (vArg
    (pi (vArg `A)
      (abs "-" (var 2
        (vArg (var 1
          (vArg (var 0 [] :: []) :: []))))))
      -- α : A → W
      -- A → P(α(a))
      -- a : A
      -- P(α(a))
      -- α ref
      -- a ref
      (abs "-" (pi (vArg `B)
        (abs "-" (pi (vArg `W)
          (abs "-" (pi (vArg (var 4
            (vArg (var 0 [] :: []) :: []))
              -- ω ref
              (abs "-" (var 5
                (vArg (con (quote g)
                  (vArg (var 4 [] ::
                    (vArg (var 2 [] ::
                      (vArg (var 1 [] :: [])) :: []))))))))))
                    -- g ref
                    -- α ref
                    -- β ref
                    -- ω ref
  ))))

```

Figure 9. Abstract syntax tree for constructor  $d'$

following the same approach as the recursion principle. The computation rule corresponding to  $f'$  is given as follows.

$$f'(P, d', g(\alpha, \beta, \omega)) \equiv d'(\alpha, f' \circ \alpha, \beta, \omega, f'(\omega))$$

We can automate the generation of the induction rule  $f'$  and its corresponding computation rules for the type  $W$  using `generateInd` interface. The following code gives the implementation of `generateInd`.

```

generateInd : Arg Name → Name → (indexList : List Nat) → TC T
generateInd (arg i f) t indLs =
  do indLs' ← getIndex t indLs
  cns ← getConstructors t
  lcons ← getLength cns
  cls ← getClauseDep lcons zero t f indLs' cns
  RTy ← getType t
  funType ← getRtypeInd t zero indLs' RTy
  declareDef (arg i f) funType
  defineFun f cls

```

`generateInd` uses `getClauseDep` to generate the clause definitions representing the computation rule of the type  $W$ . The abstract representation of the type of  $W$  is provided by `getRtypeInd`.  $f'$  generated by `generateInd` is brought into scope by `unquoteDecl` as follows.

```
unquoteDecl f' = generateInd (vArg f') (quote W) []
```

We pass an empty list to `generateInd` as the type  $W$  has no index. We can also pass an empty list if all the constructors of a type has the same number of index as the parent type. But if any one constructor has an index count different from the index count of the parent type, then we have to explicitly pass the index count of all the constructors.

## 4 Code Generation for Higher Inductive Type

In Agda, there are no built-in primitives to support the definition of higher inductive types. However, we can still define a higher inductive type with a base type using Dan Licata's [11] method as discussed in section 2.1. In this section, we

discuss the automation of code generation for the boilerplate code segments defining the higher inductive type. We also describe how to automate the code generation for the elimination and the computation rules of the higher inductive type using static type information obtained from the base type.

### 4.1 Higher Inductive Type Definition

In Agda, we define an inductive type using `data` keyword. The data literal characterize a data type by declaring its type and specifying its constructors. Consider the following generic form for the definition of an inductive type  $W$  with constructors  $g_1 \dots g_n$ .

$$\begin{aligned}
 \text{data } W^* & (x_1 : P_1) \dots (x_n : P_n) : Q_1 \rightarrow \dots \rightarrow Q_n \rightarrow \text{Set } \ell \text{ where} \\
 g_1^* & : \{i_1 : Q_1\} \dots \{i_n : Q_n\} \rightarrow \text{Type}_1 \rightarrow W^*_{x_1 \dots x_n i_1 \dots i_n} \\
 & \vdots \\
 g_n^* & : \{j_1 : Q_1\} \dots \{j_n : Q_n\} \rightarrow \text{Type}_n \rightarrow W^*_{x_1 \dots x_n j_1 \dots j_n}
 \end{aligned}$$

The parameters  $(x_1 : P_1) \dots (x_n : P_n)$  are common to all the constructors, and the type of each constructor implicitly encodes the parameter references. However, the indices are different for each constructor. So, the type of each constructor explicitly exhibits the index declaration. During the construction of a higher inductive type, we have to represent the parameters and the indices as explicit arguments in the type of the constructors. We define a higher inductive type  $W$  as a top-level definition using a base type  $W^*$  similar to the module `Circle` in section 2.1. The reflection type Definition provides us the type and the constructors of the base type  $W^*$ . We copy the type of  $W^*$  to  $W$  and for the constructors  $g_1 \dots g_n$  of  $W$ , we traverse the abstract representation of the type of  $g_1^* \dots g_n^*$  respectively replacing the occurrence of  $W^*$  to  $W$  in every strict positive position. Consider a constructor  $g_i^*$  having the following type.

$$g_i^* : (A \rightarrow W^*) \rightarrow (B \rightarrow W^*) \rightarrow C \rightarrow W^* \rightarrow W^*$$

We built the type of  $g_i$  by traversing the abstract syntax tree of  $g_i^*$  and replacing the base type  $W^*$  with the higher inductive type  $W$ . The abstract syntax tree of  $g_i^*$  incorporates the type of the parameters and the indices if present. We have to retain the parameters and the indices explicitly during the construction of  $g_i$ . The following equation represents the type of the constructor  $g_i$ .

$$g_i : (A \rightarrow W) \rightarrow (B \rightarrow W) \rightarrow C \rightarrow W \rightarrow W$$

We explicitly pass the type of the path constructors to the automation tool. The higher inductive type definition of `Circle` in section 2.1 represents the path constructors as propositional equalities. The automation tool takes the path types as input and declares them as propositional equalities using the reflection primitive `declarePostulate`. We introduce a new data type `ArgPath` to input the path types to the automation tool.



*data ArgPath*  $\{\ell_i\} : \text{Set } (\text{Isuc } \ell_i)$  where  
*argPath* :  $\text{Set } \ell_i \rightarrow \text{ArgPath}$

The constructor *argPath* takes the type of a path constructor as input. We define the generic form of a higher inductive type as follows.

```
data-hit (quote W*) W
  Wpoints (g1 :: ... :: gn :: [])
  Wpaths (p1 :: ... :: pn :: [])
  (argPath
    ({x1 : P1} → ... → {xn : Pn} →
     {i1 : Q1} → ... → {in : Qn} → Typee1 →
     (gi{x1}...{xn}{i1}...{in}...) ≡ (gj...)) ::
    ...
  )
  argPath
    ({x1 : P1} → ... → {xn : Pn} →
     {j1 : Q1} → ... → {jn : Qn} → Typeen →
     (gi{x1}...{xn}{j1}...{jn}...) ≡ (gj...)) :: []
```

We define holders *Wpoints* for point constructors and *Wpaths* for path constructors as part of the higher inductive type definition of *W*. We cannot retrieve the constructors of the higher inductive type *W* using Definition. Therefore, *Wpoints* and *Wpaths* act as the only references for the constructors of *W*. The elements of the *argPath* list represent the type of the path constructors  $p_1 \dots p_n$  respectively. We explicitly include the parameter references  $\{x_1 : P_1\} \dots \{x_n : P_n\}$  and the index references  $\{k_1 : Q_1\} \dots \{k_n : Q_n\}$  in the type of the arguments to *argPath*. The points  $g_1 \dots g_n$  are not in scope when used in the identity type passed to *argPath*. The automation tool uses the base type constructors  $g_1^* \dots g_n^*$  as dummy arguments in the place of  $g_1 \dots g_n$  respectively. The automation tool implements the interface *data-hit* as follows.

```
data-hit : ∀{ℓ1} (baseType : Name) → (indType : Name) →
  (pointHolder : Name) → (lcons : List Name) →
  (pathHolder : Name) → (lpaths : List Name) →
  (lpathTypes : (List (ArgPath {ℓ1}))) → TC T
data-hit baseType indType pointHolder lcons pathHolder lpaths lpathTypes =
  do defineHindType baseType indType
     lcons' ← getConstructors baseType
     defineHitCons baseType indType lcons' lcons
     lpathTypes' ← getPathTypes baseType indType lcons' lcons lpathTypes
     defineHitPathCons lpaths lpathTypes'
     definePointHolder pointHolder lcons
     definePathHolder pathHolder lpaths
```

The higher inductive type *W*, the points  $g_1 \dots g_n$ , the paths  $p_1 \dots p_n$ , and the holders *Wpoints* and *Wpaths* are brought into scope by *unquoteDecl*. In the above implementation of *data-hit*, *defineHindType* defines the higher inductive type as a top-level definition using the base type. *defineHitCons* specifies the point constructors of the higher inductive type using the type information obtained from the constructors of the base type, and *defineHitPathCons* builds the paths constructors of the higher inductive type

using the *argPath* list. The following code automates the generation of the higher inductive type definition for *Circle* given in section 2.1.

```
unquoteDecl S Spoints base Spaths loop =
  data-hit (quote S*) S
    Spoints (base :: []) -- point constructors
    Spaths (loop :: []) -- path constructors
    (argPath (base* ≡ base*) :: []) -- base replaces base*
```

The identity type input ( $\text{base*} \equiv \text{base*}$ ) to *argPath* represents the type of the path loop, and it uses the inductive type constructor *base\** as a dummy argument in the place of the higher inductive type constructor *base*. The constructor *base* comes into scope only during the execution of *unquoteDecl*, and so cannot be used in the identity type reference in *argPath*. We use the constructor *base\** of type *S\** as dummy argument because the type of *base\** is similar to *base*, and has the same references for the common arguments. The automation tool traverses the abstract syntax tree of *loop* and replaces the occurrences of *base\** with *base*.

## 4.2 Non-dependent Eliminator

Non-dependent eliminator or the recursion principle of a higher inductive type *W* maps the points and paths of *W* to an output type *P*. Let  $g_1 : A \rightarrow W$  and  $g_2 : B \rightarrow W$  be the point constructors, and  $l : (\alpha : A) \rightarrow (\beta : B) \rightarrow (g_1 \alpha) \equiv (g_2 \beta)$  be the path constructor of *W*. To define a mapping  $f : W \rightarrow P$ , we need points  $d_1 : A \rightarrow P$  and  $d_2 : B \rightarrow P$  and a path  $h : (\alpha : A) \rightarrow (\beta : B) \rightarrow (d_1 \alpha) \equiv (d_2 \beta)$  in the space *P*. The path constructor *l* can take arguments of any type including the higher inductive type *W* but for simplicity we will consider only arguments of constant type.

To construct the recursion principle *f*, we need to build the type of points  $d_1, d_2$  and path *h*. The type of  $d_1$  and  $d_2$  are built from the abstract syntax tree of points  $g_1$  and  $g_2$  respectively using the approach described in section (3.1). We build the type of *h* by traversing the abstract syntax tree of *l*. During the traversal, we replace the points  $g_1$  and  $g_2$ , which are arguments to the identity type, in the co-domain of the path *l* by the points  $d_1$  and  $d_2$  respectively. In figure (10), the arguments of the identity type in the abstract syntax tree of *h* references the points  $d_1$  and  $d_2$  using de Bruijn index 3 and 2 respectively. We copy the other two arguments of *l*, which are of constant type *A* and *B*, without any changes into the type of *h*.

The recursion rule *f* corresponding to figure (10) is given by the following equation.

$$f : (P : \text{Set}) \rightarrow (d_1 : A \rightarrow P) \rightarrow (d_2 : B \rightarrow P) \rightarrow \\
(h : (\alpha : A) \rightarrow (\beta : B) \rightarrow (d_1 \alpha) \equiv (d_2 \beta)) \rightarrow \\
W \rightarrow P$$

The computation rules for constructors  $g_1$  and  $g_2$  give the action of eliminator *f* on input  $g_1(\alpha)$  and  $g_2(\beta)$  respectively.



```
(def (quote _≡_)
  (vArg (def (quote ap)
    (vArg (lam visible (abs "x"
      (def (quote f)
        (vArg (var 6 []) ::
          vArg (var 5 []) ::
          vArg (var 4 []) ::
          vArg (var 3 []) ::
          vArg (var 0 []) :: [])))))) ::
    vArg (def (quote l)
      (vArg (var 1 []) ::
        vArg (var 0 []) :: [])) :: [])) ::
    vArg (var 2
      (vArg (var 1 []) ::
        vArg (var 0 []) :: [])) :: []))
```

**Figure 11.** Abstract syntax tree for the term representing the action of function  $f$  on path  $l$

The computation rules say that the function  $f$  maps the points  $g_1(\alpha)$  and  $g_2(\beta)$  to the points  $d_1(\alpha)$  and  $d_2(\beta)$  respectively in the output type  $P$ . We specify the clause definition to build the computation rules for  $g_1$  and  $g_2$  in the same way as for a regular inductive type (sec. 3.1) except that the clause definition for the higher inductive type  $W$  has an additional reference variable to the path  $h$ . The following definitional equalities give the computation rules for  $g_1$  and  $g_2$ .

The computation rule for the path constructor  $l$  defines the action of function  $f$  on the path  $(l \alpha \beta)$ . In type theory, the functions are functorial, and they preserve the path structures in their mapping to the output type [17]. The function  $ap_f$  (1) defines the action of the non-dependent eliminator  $f$  on the path  $l$ . Using (1), we can build the computation rule for the path  $l$  as follows.

The computation rule  $\beta_f$  exists only as propositional equality. We build the type of  $\beta_f$  using the same approach as for the recursion rule  $f$ . The type of  $f$  and  $\beta_f$  is identical except for the mapping  $W \rightarrow P$  in  $f$  which is replaced by the term representing the action of function  $f$  on the path  $(l \alpha \beta)$ . The function  $ap$  applies  $f$ , which is nested inside a lambda function, on the arguments  $(g_1 \alpha)$  and  $(g_2 \beta)$  of the path  $(l \alpha \beta)$ . We use the constructor `lam` of `Term` to introduce the lambda function with argument  $x$  (fig. 11). Inside the lambda

body, the argument  $x$  gets a de Bruijn index reference of zero and we update the remaining de Bruijn indices accordingly. The application of function  $f$  to the path  $(l \alpha \beta)$  substitutes the points  $(g_1 \alpha)$  and  $(g_2 \beta)$  for the lambda argument  $x$ , and it evaluates to the path  $(h \alpha \beta)$  in the output type  $P$ . We implement the `generateRecHit` interface as follows.

```

1 generateRecHit : Arg Name → List (Arg Name) →
2   (baseType : Name) → (indexList : List Nat) →
3   (baseRec : Name) → (indType : Name) →
4   (points : List Name) → (paths : List Name) → TC T
5 generateRecHit (arg i f) argD baseType indLs baseRec indType points paths =
6   do lcons ← getConstructors baseType
7     lpoints ← getLength points
8     lpaths ← getLength paths
9     clauses ← getPathClause lpoints lpaths baseRec
10    RTy ← getType baseType
11    funTypePath ← getRTypePath baseType indType baseRec indLs paths zero RTy
12    declareDef (arg i f) funTypePath
13    defineFun f clauses
14    generateβRecHit argD baseType indLs baseRec indType f points paths

```

generateRecHit takes the base type recursion rule as input and uses that to map the points  $g_1(\alpha)$  and  $g_2(\beta)$  to  $d_1(\alpha)$  and  $d_2(\beta)$  respectively in the abstract syntax tree of the path  $(l \ \alpha \ \beta)$ . The second argument `argD` is a list of terms representing the computation rules for the path constructors. The `generateβRecHit` interface takes `argD` as input and builds the computation rule for the path constructor  $l$ . Other inputs to `generateRecHit` are the point and path holders declared during the higher inductive type definition of  $W$ . The following automation code generates the recursion rule and the computation rules for `Circle`.

```
unquoteDecl recS1* = generateRec (vArg recS1*)
                        (quote S1*) (0 :: [])

unquoteDecl recS1 βrecS1 = generateRecHit (vArg recS1)
                        ((vArg βrecS1) :: [])
                        (quote S1*) (0 :: [])
                        (quote recS1*)
```

```

(pi (vArg (`W `=> agda-sort (lit 0)))
  (abs "P")
  (pi (vArg
    (pi (vArg `A)
      (abs "α")
      (var 1 (vArg (con (quote g1)
        (vArg (var 0 [])) :: [])) :: []))))))
    (abs "d1"
    (pi (vArg
      (pi (vArg `B)
        (abs "β")
        (var 2 (vArg (con (quote g2)
          (vArg (var 0 [])) :: [])) :: []))))))
        (abs "d2"
        (pi (vArg
          (pi (vArg `A)
            (abs "α")
            (pi (vArg `B)
              (abs "β")
              (def (quote _≡_)
                (vArg (def (quote transport)
                  (vArg (var 4 [])) ::
                    vArg (def (quote l) (vArg (var 1 [])) ::
                      vArg (var 0 [])) :: [])) ::
                    vArg (var 3 (vArg (var 1 [])) :: [])) :: [])) ::
                    vArg (var 2 (vArg (var 0 [])) :: [])) :: []))))))
          (abs "h"
          (pi (vArg `W)
            (abs "ω" (var 4 (vArg (var 0 [])) :: [])))))))))

```

Figure 12. Abstract syntax tree of function  $f'$ 

(quote  $S_1$ )  $S_1$ points  $S_1$ paths

The term  $\beta_{\text{rec}S_1}$  represents the computation rule for the path constructor loop and is brought into scope by `unquoteDecl`.

### 4.3 Dependent Eliminator

Dependent eliminator or the induction principle of a higher inductive type  $W$  is a dependent function that maps an element  $\omega$  of  $W$  to an output type  $P \omega$ . For the type  $W$  with point constructors  $g_1 : A \rightarrow W$  and  $g_2 : B \rightarrow W$  and path constructor  $l : (\alpha \rightarrow A) \rightarrow (\beta \rightarrow B) \rightarrow (g_1 \alpha) \equiv (g_2 \beta)$ , to define a mapping  $f' : (\omega : W) \rightarrow P \omega$ , we need  $d_1 : (\alpha : A) \rightarrow P \alpha$ ,  $d_2 : (\beta : B) \rightarrow P \beta$ , and  $h : (\alpha \rightarrow A) \rightarrow (\beta \rightarrow B) \rightarrow \text{transport } P (l \alpha \beta) (d_1 \alpha) \equiv (d_2 \beta)$ , where  $(h \alpha \beta)$  is a heterogeneous path transported over  $(l \alpha \beta)$  lying in the space  $P \beta$ . Equation (3) gives the type of  $\text{transport}$ .

In the abstract syntax tree of  $f'$  (fig. 12), the type of constructors  $d_1$  and  $d_2$  are built from the type of  $g_1$  and  $g_2$  respectively using the same approach as in section (3.2). When building the type of  $d_1$ , we copy the constant argument  $A$  directly into the abstract syntax tree of  $d_1$ , and for the codomain  $P$ , which depends on the action of constructor  $g_1$  on the constant argument  $\alpha : A$ , we pass the de Bruijn index reference of  $\alpha$  as an argument to the constructor  $g_1$  built using the reflection construct `con`. Similarly, for  $d_2$ , we copy the type  $B$  directly into the abstract syntax tree of  $d_2$  and

pass the de Bruijn index of  $\beta : B$  to the constructor  $g_2$  in the co-domain. We build the type of path  $h$  by traversing the abstract syntax tree of  $l$  and adding relevant type information into it. We copy the constant arguments of type  $A$  and  $B$  without modification. For the codomain of  $l$ , which is the identity type  $(g_1 \alpha) \equiv (g_2 \beta)$ , we insert the quoted name of function *transport* with arguments  $P$ , the path  $(l \alpha \beta)$  and  $(d_1 \alpha)$ . We apply the base eliminator to the arguments of the path  $(l \alpha \beta)$  to form the elements  $(d_1 \alpha)$  and  $(d_2 \beta)$ . The following declaration gives the type of  $f'$  corresponding to figure (12).

$$\begin{aligned}
 f' : (P : W \rightarrow \text{Set}) \rightarrow \\
 (d_1 : (\alpha : A) \rightarrow P(g_1 \alpha)) \rightarrow (d_2 : (\beta : B) \rightarrow P(g_2 \beta)) \rightarrow \\
 (h : (\alpha : A) \rightarrow (\beta : B) \rightarrow \\
 \text{transport } P (l \alpha \beta) (d_1 \alpha) \equiv (d_2 \beta)) \rightarrow \\
 (\omega : W) \rightarrow P(\omega)
 \end{aligned}$$

We build the computation rules corresponding to the mapping  $f'$  for the point constructors  $g_1$  and  $g_2$  using the same approach as for the non-dependent eliminator  $f$ . The following equations give the computation rules for the point constructors  $g_1$  and  $g_2$ .

$$\begin{aligned}
 f'(P, d_1, d_2, h, g_1(\alpha)) &\equiv d_1(\alpha) \\
 f'(P, d_1, d_2, h, g_2(\beta)) &\equiv d_2(\beta)
 \end{aligned}$$

The computation rule  $\beta_{f'}$  for the heterogeneous path  $h$  depends on *apd* (2) which gives the action of dependent function  $f'$  on the path  $h$ . We build the type of  $\beta_{f'}$  using (2) as follows.


$$\begin{aligned}
 \beta_{f'} : (P : W \rightarrow \text{Set}) \rightarrow \\
 (d_1 : (\alpha : A) \rightarrow P(g_1 \alpha)) \rightarrow (d_2 : (\beta : B) \rightarrow P(g_2 \beta)) \rightarrow \\
 (h : (\alpha : A) \rightarrow (\beta : B) \rightarrow \\
 \text{transport } P (l \alpha \beta) (d_1 \alpha) \equiv (d_2 \beta)) \rightarrow \\
 \text{apd}_{(\lambda x \rightarrow f'(P, d_1, d_2, h, x))} (l \alpha \beta) \equiv (h \alpha \beta)
 \end{aligned}$$

We build the abstract syntax tree for the computation rule  $\beta_{f'}$  in the same way as for induction principle  $f'$ . In the codomain of  $\beta_{f'}$  (fig. 13), which is an identity type, we insert the abstract representation of the function *apd* with arguments  $f'$ , nested inside a lambda body, and the path  $(l \alpha \beta)$ . We pass the path  $(h \alpha \beta)$ , built using de Bruijn indices, as the second argument to the identity type. The following code gives the implementation of the `generateIndHit` interface in the automation tool.

```

generateIndHit : Arg Name → List (Arg Name) →
  (baseType : Name) → (indLs : List Nat) →
  (baseElm : Name) → (indType : Name) →
  (points : List Name) → (paths : List Name) → TC T
generateIndHit (arg i f) argD baseType indLs baseRec indType points paths =
  do indLs' ← getIndex baseType indLs
  lcons ← getConstructors baseType
  lpoints ← getLength points
  lpaths ← getLength paths
  clauses ← getPathClauseDep lpoints lpaths baseType baseRec indLs' lcons
  RTy ← getType baseType

```



**Figure 13.** Abstract syntax tree for the term representing the action of dependent function  $f'$  on path  $l$

```
funTypePath ← getRtypePathDep baseType indType baseRec points paths zero indLs' RType
declareDef (arg i f) funTypePath
defineFun f clauses
generateβIndHit argD baseType indLs baseRec indType f points paths
```

The interface `generateIndHit` takes the base eliminator as an argument and uses that to map the point constructors  $g_1$  and  $g_2$  to output  $d_1$  and  $d_2$  when building the codomain of the path constructor  $h$ . `generate $\beta$ IndHit` builds the computation rule for the path constructor  $h$ . The following code builds the induction principle and the computation rules for `Circle`.

```
unquoteDecl indS1* = generateInd (vArg indS1*)
                        (quote S1*) []

unquoteDecl indS1 βindS1 = generateIndHit (vArg indS1)
                        ((vArg βindS1) :: [])
                        (quote S1*) []
                        (quote indS1*)
                        (quote S1) S1points S1paths
```

The primitive `unquoteDecl` declares  $\beta\text{ind}_{S_1}$  as a postulate.  $\beta\text{ind}_{S_1}$  gives the action of dependent function  $\text{ind}_{S_1}$  on the path constructor `loop`.

## 5 Application

The field of homotopy type theory is less well-developed on the programming side. There are only few programming applications of homotopy type theory, and the role of computationally relevant equality proofs on programming is an area of active research. Applications such as homotopical patch theory [4] discuss the implementation of Darcs [16] version control system using patch theory [12] [9] in the context of homotopy type theory. Containers in homotopy type theory [3] [2] implement data structures such as multisets and cycles. The automation tool discussed in this paper abstracts away the difficulties involved in the implementation of a higher inductive type and its elimination rules. It introduces interfaces which simplify the intricacies of a

higher inductive type definition and usage by automating the generation of the code segments defining the higher inductive type and its elimination rules. The automation tool is significant in reducing the development effort for existing applications, and it can also attract new programming applications in homotopy type theory.

In the next section, we discuss the implementation of patch theory application in Agda and exhibit a tremendous reduction in code size using the automation tool. In section 5.2, we present a cryptography application and discuss how to abstract the implementation difficulties of a higher inductive type making it more accessible to the cryptographers.

## 5.1 Patch Theory Revisited

A patch is a syntactic representation of a function that modifies a repository context when applied. For example, a patch  $(s1 \leftrightarrow s2 @ l)$ , which replaces string  $s1$  with  $s2$  at line  $l$ , when applied to a repository context with string  $s1$  at line  $l$  results in a repository context with string  $s2$  at line  $l$ . In homotopical patch theory [4], the patches are modeled as paths in a higher inductive type. The higher inductive type representation of patches automatically satisfy groupoid laws such as the composition of patches is associative, and inverse composes to identity. Domain-specific laws related to the patches such as two swaps at independent lines commute are designed as higher dimensional paths. The computation content of the patches is extracted by mapping them to bijections in the universe with the help of univalence. Due to the functoriality of mappings in type theory, the functions preserve the path structures in their mapping to the universe.

We developed the patch theory application in Agda using Dan Licata’s method [11]. We implemented basic patches like the insertion of a string as line l1 in a file and deletion of a line l2 from a file. The functions implementing insertion and deletion in the universe are not bijective. So, to map the paths representing the patches insert and delete into the universe, we used the patch history approach [4]. According to this approach, we developed a separate higher inductive type *History* which serves as the types of patches. In addition to basic patches, we also implemented patches of encryption using cryptosystems like rsa [15] and paillier [13].

We used the automation tool described in this paper to generate code for the higher inductive type definition representing *History* and the repository context *cryptR* for the patches. We also automated the code generation for the elimination and the computation rules for the higher inductive types *History* and *cryptR*. In addition to abstracting the implementation difficulties of higher inductive types, the automation tool helped us to achieve an extensive reduction in the code size of the original application. We were able to automate the generation of approximately 1500 lines of code with just about 70 lines of automation code. The automation massively reduced the code size of the application which is

about 2500 lines resulting in 60% reduction in the original code size.

## 5.2 Cryptography

The work of [18] applies the tools of homotopy type theory for cryptographic protocol implementation. It introduces a new approach for the formal specification of cryptographic schemes using types. The work discusses modeling *cryptDB* [14] using a framework similar to patch theory. *CryptDB* employs layered encryption techniques and demonstrates computation on top of encrypted data. We can implement *cryptDB* by modeling the database queries as paths in a higher inductive type and mapping the paths to the universe using singleton types [4]. The automation tool can be applied to generate code for the higher inductive type representing *cryptDB* and its corresponding elimination and computation rules. By using the automation tool, we can abstract the convolutions of homotopy type theory thus making it more accessible to the broad community of cryptography.

A formal specification of a cryptographic construction promises correctness of properties related to security and implementation. The downside of formal specification is that it introduces a framework which requires expert knowledge on theorem proving and a strong mathematical background. By automating the code constructions for the mathematical part such as the higher inductive type implementation, we simplify theorem proving and formal specification to a considerable extent and make it more accessible to regular programmers without a strong mathematical background.

## 6 Conclusion and Future Work

We presented an automation tool developed using the new reflection library of Agda extended with support for elaborator reflection. Our automation tool handles code generation for inductive types with constructors taking zero arguments, one or more arguments, and type being defined itself as an argument. We simplified the syntax for defining higher inductive types through the mechanized construction of the boiler-plate code segments. By automating the generation of the elimination and the computation rules associated with a higher inductive type, we demonstrated an extensive reduction in code size and abstraction of difficulties involved in implementing and using the higher inductive type. Next, we intend to extend the support to include more categories of the inductive type such as the inductive-inductive type and the inductive-recursive type.

## References

- [1] 2017. *Agda's Documentation*. <http://agda.readthedocs.io/en/latest/language/reflection.html>.
- [2] Michael Abbott, Thorsten Altenkirch, and Neil Ghani. 2005. Containers: constructing strictly positive types. *Theoretic Computer Science*.
- [3] Thorsten Altenkirch. 2014. Containers in homotopy type theory. (January 2014). Talk at Mathematical Structures of Computation, Lyon.
- [4] Carlo Angiuli, Edward Morehouse, Daniel R. Licata, and Robert Harper. 2014. Homotopical Patch Theory. In *Proceedings of the 19th ACM SIGPLAN International Conference on Functional Programming (ICFP '14)*. ACM.
- [5] Ana Bove and Venanzio Capretta. 2005. Modelling general recursion in type theory. *Mathematical Structures in Computer Science*, 15(4):671–708.
- [6] David Christiansen. 2005. Dependent type providers. In *Proceedings of the 9th ACM SIGPLAN Workshop on Generic Programming, WGP '13*, New York, USA.
- [7] David Christiansen. 2016. *Practical Reflection and Metaprogramming for Dependent Types*. Ph.D. Dissertation. IT University of Copenhagen.
- [8] David Christiansen and Edwin Brady. 2016. Elaborator Reflection: Extending Idris in Idris. In *Proceedings of the 21st ACM SIGPLAN International Conference on Functional Programming (ICFP '16)*. ACM, Nara, Japan.
- [9] Jason Dagit. 2009. Type-correct changes—a safe approach to version control implementation. (2009). MS Thesis.
- [10] Gabriel Ebner, Sebastian Ullrich, Jared Roesch, Jeremy Avigad, and Leonardo de Moura. 2017. A Metaprogramming Framework for Formal Verification. *Proceedings of the ACM on Programming Languages* 1, ICFP, Article 34 (August 2017), 29 pages.
- [11] Daniel R. Licata. 2011. Running Circles Around (In) Your Proof Assistant; or, Quotients that Compute. (April 2011). <https://homotopytypetheory.org/2011/04/23/running-circles-around-in-your-proof-assistant>.
- [12] Samuel Mimram and Cinzia Di Giusto. 2013. A categorical theory of patches. *Electronic Notes in Theoretic Computer Science*, 298:283–307.
- [13] Pascal Paillier. 1999. Public-key cryptosystems based on composite degree residuosity classes. In: *Proceedings of the 18th Annual International Conference on the Theory and Applications of Cryptographic Techniques (EUROCRYPT)*, Prague, Czech Republic.
- [14] Raluca Ada Popa, Catherine M.S. Redfield, and Hari Balakrishnan Nickolai Zeldovich. 2011. *CryptDB: Protecting confidentiality with encrypted query processing*. In: *Proceedings of the 23rd ACM Symposium on Operating Systems Principles (SOSP)*, Cascais, Portugal.
- [15] Adi Shamir Ron Rivest and Leonard Adleman. 1978. A Method for Obtaining Digital Signatures and Public-Key Cryptosystems. *Communications of the ACM*, 21 (2), pp. 120-126.
- [16] David Roundy. 2005. Darcs: Distributed version management in haskell. In *ACM SIGPLAN Workshop on Haskell*. ACM.
- [17] The Univalent Foundations Program. 2013. *Homotopy Type Theory: Univalent Foundations of Mathematics*. <https://homotopytypetheory.org/book>, Institute for Advanced Study.
- [18] Paventhan Vivekanandan. 2018. A Homotopical Approach to Cryptography. (July 2018). To be presented at Workshop on Foundations of Computer Security (FCS 2018), University of Oxford, UK.